
Overview 
This session gives an overview of COM+ services, beginning with the basics of COM 
and how component design changes under COM+. It then moves into creating COM+ 
components and how to administer COM+ applications and security. It then moves onto 
transaction management and transacting VFP tables using Compensating Resource 
Managers. Asynchronous events such as Queued Components and Loosely Coupled 
Events are covered.  

http://www.craigberntson.com

Reviewing COM 
For many years, COM has been the method used to integrate and automate 
applications under Windows. For example, if you automate Word, Excel, or Outlook 
from your Visual FoxPro application, you do it through COM.  

COM is an acronym for the Component Object Model. It is a specification for writing 
reusable software that runs in component-based systems. It turns out there are two 
types of COM components, in-process and out-of-process.  

An in-process component is compiled as a DLL and requires an EXE to host it. When 
you call the component, it is hosted by your application EXE, thus it runs in your 
application s memory space on the same computer. An out-of-process component is 
compiled as an EXE and runs in its own memory space. There are pros and cons of 
each method. If the in-process DLL should crash, it could corrupt the memory space of 
your EXE and cause it to crash too. This isn t the case with a COM EXE. Because it 
runs in its own memory space, it is considered safer to run. However, the data transfer 
across the process boundaries from your EXE to the COM EXE takes longer than from 
your EXE to the COM DLL, making the EXE slower.  

The application that uses COM component is called the client and the COM component 
is called the server. Do not confuse this with Client/Server databases. Visual FoxPro 
provides the capability to create clients and both in-process and out-of-process servers, 
with one caveat. VFP in-process components do not support user interfaces, so you 
can t get any screen output from them.  

Whether you compile an in-process DLL or an out-of-process EXE, what happens under 
the hood is pretty much the same. You application never directly talks to the COM 
component. Instead, it talks to a proxy. The data is communicated across the process 
boundary to a stub, which then passes communication on to the COM component 
(Figure 1). You don t have to worry about creating the proxy and stub or a 
communication channel for the data. Windows does all that for you. The work that 
Windows does for this is frequently called plumbing (after the plumbing in your house).  

Figure 1. COM components talk to each other across a process boundary. 

When you use a COM component, you typically connect to it, do lots of work with it, 
then disconnect at the very end. This is a stateful connection, meaning the component 
exists (has state) for a lengthy period of time. 

Creating a COM component 
At this point, you may be wondering how to create a COM component. The following 
steps walk you through creating a COM component. 
Create a new project called BizRules 

Add a new program file named MathClass.prg. Enter the code in Listing 1 into the editor. 

Listing 1. This code is used to create a COM component 

DEFINE CLASS Math AS SESSION OLEPUBLIC 
  nNumber1 = 0 
  nNumber2 = 0 

 

  FUNCTION Add(tnOne AS Integer, tnTwo AS Integer) AS Integer 
    LOCAL lnResult 

 

    WITH This 
      IF PARAMETERS() = 2 
        .nNumber1 = tnOne 
        .nNumber2 = tnTwo 
      ENDIF 

 

      lnResult = .nNumber1 + .nNumber2 
    ENDWITH 

    RETURN lnResult 
  ENDFUNC 
   
  FUNCTION Multiply(tnOne AS Integer, tnTwo AS Integer) AS Integer 
    LOCAL lnResult 

 
    WITH This 
      IF PARAMETERS() = 2 
        .nNumber1 = tnOne 
        .nNumber2 = tnTwo 
      ENDIF 

 

      lnResult = .nNumber1 + .nNumber2 

 

   ENDWITH 
    RETURN lnResult 
  ENDFUNC 
ENDDEFINE 

Save and close the program and select Build from the Project Manager. The Build Options dialog is 
displayed 

Select Single-threaded COM server (DLL) then click OK. Visual FoxPro compiles the program and 
creates the DLL.  

Notice the OLEPUBLIC keyword. This tells the compiler to create a COM component. If 
you design the class visually, check OLE Public on the Class Info dialog (Figure 2).  

 



Using Windows Component Services (COM+) with Visual FoxPro 

© Craig Berntson, 2005         
Page 6 of 49 

Figure 2. The Class Info dialog is used to set information for the entire class. 

The Build Option dialog gave three choices for building a COM component, as listed in 
Table 1. 
Table 1. The Build dialog contains three options for creating a COM component 
Build Action Description 
Win32 Executable / COM server (EXE) Creates an out-of-process EXE COM server 
Single-threaded COM server (DLL) Creates a single-threaded in-process DLL COM server 
Multi-threaded COM server (DLL) Creates a multi-threaded in-process DLL COM server 

 

A single-threaded COM server can only be instantiated one time. Subsequent calls will 
be blocked and will wait until the COM server is available. For example, if you have a 
COM server that runs a lengthy process and multiple users attempt to call it at the same 
time, the first user will get access and other users will wait their turn. A single-threaded 
COM server is useful when the component is installed locally on the workstation.  

A multi-threaded COM server gets around this problem. Multiple calls are not blocked. 
Use multi-threaded COM server when the component should be installed on a server for 
multi-user access. This is the type of COM DLL you will compile when creating COM+ 
components.  

When VFP builds a COM component, it generates several registry entries that point to 
the directory and file that was compiled. This way, when the component is used, VFP 
can find it when it s run on the development machine. In addition, a type library is 
created. This file contains the same root name as the DLL but with a .tlb extension. If 
you distribute the COM DLL, you need to include the type library and the proper VFP 
runtime libraries.  

Now that you ve created the COM component, it s time to use it. The following steps 
show you how to do this. 
In the Command Window enter MODIFY COMMAND TestLocal  

Enter the code in listing 2. 

Save and run the program. 

Listing 2. Code to instantiate and use a COM component. 

CLEAR 
ox = CREATEOBJECT( BizRules.Math ) 
? ox.Multiply(2, 3) 
ox.nNumber1 = 5 
ox.nNumber2 = 6 
? ox.Add() 

When you run the program, first the screen is cleared, the COM component is 
instantiated. VFP looks up BizRules.Math in the registry to get the location of the COM 
file. Then, the Multiply method is called and the result of 6 is displayed. The two 

numbers are then set to new values, the Add method is called, and the result of 11 is 
displayed. When the program ends, the component is released. 

What is DCOM? 
Under the Distributed Component Object Model (DCOM), the component is installed on 
a remote computer, typically a server. The component runs in memory on the server, 
rather than the work station. An in-process DLL still needs to be hosted by an EXE. In 
the case of DCOM, this is typically DLLHost.exe.   

The plumbing for a DCOM application is very similar to that of a COM application, 
except that the communication is across a computer boundary (Figure 3). Windows still 
provides all the plumbing for this to happen. However, you have to do some 
configuration so the workstation knows on which server the component is installed. You 
have probably already guessed that DCOM is slower than COM because the 
communication from the application to the component needs to go across the network.  

 

Figure 3. DCOM components communicate across computer boundaries. 

One advantage of DCOM over COM is that the component is centrally located on the 
server. You don t need to distribute the component to each user when it is updated. For 
example, you can create a component that contains business rules for generating 
invoices. If those rules change, you need only to install a new DLL on the server without 
touching the client workstations. DCOM applications are also typically stateful. 

So, what are Component Services? 
Windows Component Services, also called COM+, is a set of services that host COM 
components for DCOM-type communication. At a basic level, COM+ works the same 
way as DCOM (See Figure 3). However, unlike DCOM, COM+ applications get much of 
their scalability and extensibility from stateless connections. When using stateless 
connections, you should connect to the component, run the method, then disconnect.   

In addition to increased scalability, there are a number of additional services provided 
by COM+. These services include:  

 

Administration 

 

Security 

 

Transaction management 

 

Just-in-time Activation 

 

Object Request Broker (ORB) 

 

Connection Pooling 

 

Loosely Coupled Events 

 

Queued Components 

 

Dynamic Load Balancing  

One other difference between DCOM and COM+ is that you should only use in-process 
components in COM+. These DLLs are hosted by the COM+ runtime. Before learning 
how to create a COM+ component, you need to understand a couple of other concepts, 
context and threading.  

The COM+ runtime is capable of hosting multiple instances of the same or different 
COM DLLs. Each DLL is isolated of each other by running each one in its own context. 
The DLL has access to the context to learn about the environment where it is running.   

Threading refers to how many operations at a time the application or component is 
capable of running. There are many types of threading, but I will only discuss three 
here, single, multi, and apartment-model.  

Single threading means that an application or component can only do one thing at a 
time. Visual FoxPro is mostly single threaded. It uses multiple threads internally for 
some things, but it does not give you the capability to create more than one thread. 
When you build a Win32 Executable / COM Server (exe) or a Single-threaded COM 
server (dll), you get single threading.  

When an application is multi-threaded, it is capable of doing several things at the same 
time. A VFP COM DLL that is compiled using the Multi-threaded COM server (dll) uses 
a special type of multi-threading called apartment-model threading. Under apartment 
model threading, each thread lives in its own apartment and is unaware of other 
apartments. Conflicts with global data are avoided by giving each apartment its own 
copy of the global data. You don t have to worry about apartment model threading as it 
happens automatically with VFP COM DLLs. 

Creating a COM+ Component 
Now it s time to look at how to create a COM+ component. Technically, you could use 
the DLL created earlier, but it will have no connection to its context. The code in Listing 
3 builds on the DLL by adding some code to interact with its context. 

1. Create a new project called FoxCOMPlus 

Add a new program file named MathClass.prg. Enter the code in Listing 3 into the editor.  

Listing 3. A simple COM+ component 

DEFINE CLASS Math AS SESSION OLEPUBLIC 
  nNumber1 = 0 
  nNumber2 = 0 

 

  FUNCTION Add(tnOne AS Integer, tnTwo AS Integer) AS Integer 
    LOCAL loMtx, loContext, lnResult 

 

    loMtx = CREATEOBJECT("MTXAS.APPSERVER.1") 
    loContext = loMtx.GetObjectContext() 

 

    WITH This 
      IF PARAMETERS() = 2 
        .nNumber1 = tnOne 
        .nNumber2 = tnTwo 
      ENDIF 

 

      lnResult = .nNumber1 + .nNumber2 
    ENDWITH 

 

    loContext.SetComplete() 
    loContext = NULL 
    loMtx = NULL 

 

    RETURN lnResult 
  ENDFUNC 
   
  ***************** 

 

  FUNCTION Multiply(tnOne AS Integer, tnTwo AS Integer) AS Integer 
    LOCAL loMtx, loContext, lnResult 

 

    loMtx = CREATEOBJECT("MTXAS.APPSERVER.1") 

 

   loContext = loMtx.GetObjectContext() 

 

    WITH This 
      IF PARAMETERS() = 2 
        .nNumber1 = tnOne 
        .nNumber2 = tnTwo 
      ENDIF 

 

      lnResult = .nNumber1 * .nNumber2 
    ENDWITH 

 

    loContext.SetComplete() 
    loContext = NULL 
    loMtx = NULL 

    RETURN lnResult 
  ENDFUNC 
ENDDEFINE 

Save and close the program and select Build from the Project Manager. The Build Options dialog is 
displayed. 

Select Multi-threaded COM server (DLL) then click OK. Visual FoxPro compiles the program and creates 
the DLL.  

Before you can use the new component, you need to install it in the COM+ runtime. You 
will want to install it locally for testing. For deployment, you will need to place the DLL, 
the type library, and the VFP runtime files on the server. 

Installing the component in the COM+ runtime 
COM+ organizes components into applications. The following steps walk you through 
creating the COM+ and application installing the component into the new application. 

1. From Windows Administrative Tools, select Component Services. The Component Services 
Manger is displayed (Figure 4). 

Figure 4. The Component Services Manager is used to administer COM+ applications. 

Expand the Component Services Node to Computers -> My Computer -> COM+ Applications. 

Right-click on COM+ Applications and select New -> Application from the context menu. The COM+ 
Application Install Wizard is displayed.  

Click Next to move past the welcome page. 

On the Install or Create page, you want to create a new empty application (Figure 5). 

Figure 5. Add a new COM+ application with the Install Wizard 

On the Create Empty Application page (Figure 6), enter the name for your application. For this example, 
use My COM+ App. Make sure Server Application is selected, then click Enter. 

Figure 6. When you create an empty application, you need to name the application. 

The next page, Set Application Identity (Figure 7), allows you to set the Windows user identity that this 
component runs under. The default is Interactive User. It is not a good idea to set the component to run 
under this identity as someone needs to be logged on something not common for a server. It is 
recommended that you setup a Windows user specifically for your component and assign the specific 
rights to that user that will be needed. For the purpose of this example, select Interactive User, then click 
Next. The Thank You page is displayed.  

Click Finish to create the COM+ Application. Now you need to add the FoxCOMPlus.dll to the new COM+ 
application. 

Expand the My COM+ App node in the Component Services Manager. 

Right click on Components and select New -> Component from the context menu. The Welcome page of 
the COM+ Component Install Wizard is displayed. Click Next. 

On the Import or install a component page (Figure 8), click Install new component(s). A file picker dialog 
is displayed. 

Figure 7. COM+ applications can be set to run under a specific user account.  

Figure 8. After creating a new COM+ application you need to add the components. 

Navigate to the proper folder and select both FoxCOMPlus.dll and FoxCOMPlus.tlb. The Install new 
components page is displayed (Figure 9). Click Next. The Thank You page is displayed. 



Using Windows Component Services (COM+) with Visual FoxPro 

© Craig Berntson, 2005         
Page 16 of 49 

 

Figure 9. Use the Install new components page to specify the DLLs to add to the application. 

Click Finish to complete the installation.  

To test the component, run the same test program that you ran earlier (Listing 3). To 
verify that you actually instantiate the COM+ component, select the My COM+ App -> 
Components node in the Component Services Manager (Figure 10). When the ball in 
the right-hand panel is spinning, the component is currently instantiated. 

Figure 10. After adding a component you can manage it using the Component Services Manager. 

Just In Time Activation and Statelessness 
When you instantiate the component, COM+ uses Just In Time Activation (JITA) to load 
and create the component. The CREATEOBJECT() function never really instantiates 
the component. Instead, it connects to the COM+ runtime, which in turn, instantiates the 
component, through the use of a proxy. When SetComplete() or SetAbort() is called, 
COM+ releases the component, but the proxy remains. Your application does not know 
that the component no longer exists. The COM+ runtime activates the component on 
the next call. In fact, the Init method of the component will be called again because the 
component has been instantiated again. You may notice that the second call to the 
component takes less time. That is because JITA keeps a copy of the component in 
memory for a period of time after you re done using it. The default time is 15 minutes, 
but can be changed on the Pooling & Recycling page (Figure 11) of the application s 
Property dialog in the Component Services Manager. 

